

# FootOntologyPlus

The Authors

# Indice

<b>1</b>	<b>Introduzione</b>	<b>2</b>
<b>2</b>	<b>Ontologia di partenza</b>	<b>3</b>
<b>3</b>	<b>Modifiche all'ontologia</b>	<b>4</b>
<b>4</b>	<b>Inserimento di individui</b>	<b>5</b>
<b>5</b>	<b>Regole SWRL</b>	<b>6</b>
5.1	HomeFormationRule . . . . .	6
5.2	AwayFormationRule . . . . .	6
5.3	DuplicateSubstituteRule . . . . .	7
5.4	DuplicateSubstitutedRule . . . . .	7
5.5	StarterAsSubstituteRule . . . . .	7
5.6	FriendlyMatchInTournamentRule . . . . .	7
5.7	GoalScoredByTeamRule . . . . .	7
5.8	InconsistentBallPossessionInMatchRule . . . . .	8
5.9	InconsistentContractDatesRule . . . . .	8
5.10	InconsistentMatchScoresHome/AwayRule . . . . .	8
5.11	InvalidLeagueMatchFormatExtraTime/PenaltyRule . . . . .	8
5.12	InvalidMatchRule . . . . .	9
5.13	MultipleRedCardsRule . . . . .	9
5.14	NonGoalKeeperWithSavesRule . . . . .	9
5.15	PlayerInMatchRule . . . . .	9
5.16	TeamNotInMatchRule . . . . .	9
5.17	UncontractedPlayerInMatchRule . . . . .	10
<b>6</b>	<b>Interrogazioni SPARQL</b>	<b>11</b>
<b>7</b>	<b>Conclusioni</b>	<b>12</b>

# Capitolo 1

## Introduzione

Citazione [1]

## Capitolo 2

# Ontologia di partenza

## Capitolo 3

# Modifiche all'ontologia

## Capitolo 4

### Inserimento di individui

## Capitolo 5

# Regole SWRL

**SWRL (Semantic Web Rule Language)** è un linguaggio usato nell'ambito del Semantic Web per esprimere regole di inferenza e vincoli logici più complessi di quelli realizzabili usando il solo OWL.

Nella nostra ontologia abbiamo inserito numerose regole, utili a:

1. inferire nuovi collegamenti tra i dati
2. individuare casi di errore sui dati

Le sezioni che seguono riportano tali regole, non in un ordine particolare.

### 5.1 HomeFormationRule

Questa regola "raffina" il collegamento tra una formazione e una partita, esplicitando che la prima è usata dalla squadra in casa. Questo perché `isAwayFormationIn` è una sottoproprietà di `isFormationIn`.

```
isHomeTeam(?t, ?m) ^ hasFormation(?t, ?f) ^ isFormationIn(?f, ?m) ->
    isHomeFormationIn(?f, ?m)
```

### 5.2 AwayFormationRule

Questa regola funziona in modo analogo alla precedente, ma riguardo la squadra in trasferta.

```
isAwayTeam(?t, ?m) ^ hasFormation(?t, ?f) ^ isFormationIn(?f, ?m) ->
    isAwayFormationIn(?f, ?m)
```

### 5.3 DuplicateSubstituteRule

Questa regola verifica se un giocatore segnato come riserva partecipa a multiple sostituzioni come giocatore entrante, e in tale caso gli assegna una classe di errore.

```
hasReservePlayer(?f, ?p) ^ hasEnteringPlayer(?s1, ?p) ^  
  hasEnteringPlayer(?s2, ?p) ^ differentFrom(?s1, ?s2) ^  
  hasTeamFormation(?m, ?f) ^ hasSubstitution(?m, ?s1) ^  
  hasSubstitution(?m, ?s2) -> DuplicateSubstituteError(?p)
```

### 5.4 DuplicateSubstitutedRule

Il funzionamento di questa regola è analogo alla precedente, ma riguardante il giocatore uscente. Questa regola copre però solo i giocatori titolari (ovvero i giocatori che partecipano alla partita dall'inizio).

```
hasStarterPlayer(?f, ?p) ^ hasExitingPlayer(?s1, ?p) ^  
  hasExitingPlayer(?s2, ?p) ^ differentFrom(?s1, ?s2) ^  
  hasTeamFormation(?m, ?f) ^ hasSubstitution(?m, ?s1) ^  
  hasSubstitution(?m, ?s2) -> DuplicateSubstitutedError(?p)
```

### 5.5 StarterAsSubstituteRule

Questa regola verifica se un giocatore titolare è il giocatore entrante in una sostituzione, e in tale caso gli assegna una classe di errore.

```
hasStarterPlayer(?f, ?p) ^ hasEnteringPlayer(?s, ?p) ^  
  hasTeamFormation(?m, ?t) ^ hasSubstitution(?m, ?s) ->  
  StarterAsSubstituteError(?p)
```

### 5.6 FriendlyMatchInTournamentRule

Non è corretto assegnare una partita amichevole ad un torneo, qualsiasi sia il suo tipo, e se ciò viene fatto allora una classe di errore è assegnata alla partita.

```
includedInTournament(?m, ?t) ^ FriendlyMatch(?m) ->  
  FriendlyMatchInTournamentError(?m)
```

### 5.7 GoalScoredByTeamRule

Questa regola collega un goal ad una squadra, sapendo che il goal è stato fatto da un giocatore che giocava in tale squadra nella partita.

```
scores(?p, ?g) ^ hasEvent(?m, ?g) ^ isPlayerInFormation(?p, ?f) ^  
  hasFormation(?t, ?f) ^ competesIn(?t, ?m) -> scoredByTeam(?g, ?t)
```



## 5.8 InconsistentBallPossessionInMatchRule

Questa regola controlla se la somma del possesso palla indicata nelle statistiche di performance per le squadre che hanno giocato una partita non è 100%, e in tale caso assegna una classe di errore alla partita e alle statistiche.

```
teamStatsIn(?ps1, ?m) ^ teamStatsIn(?ps2, ?m) ^ differentFrom(?ps1,
    ?ps2) ^ BallPossession(?ps1, ?p1) ^ BallPossession(?ps2, ?p2) ^
    swrlb:add(?r, ?p1, ?p2) ^ swrlb:notEqual(?r, 100) ->
    InconsistentBallPossessionInMatchError(?m) ^
    InconsistentBallPossessionInMatchError(?ps1) ^
    InconsistentBallPossessionInMatchError(?ps2)
```

## 5.9 InconsistentContractDatesRule

Questa regola verifica se le date di un contratto sono inconsistenti (data di inizio futura alla data di fine).

```
ContractStartDate(?c, ?sd) ^ ContractEndDate(?c, ?ed) ^
    temporal:before(?ed, ?sd) -> InconsistentContractDatesError(?c)
```

## 5.10 InconsistentMatchScoresHome/AwayRule

Queste due regole verificano se il vincitore di una partita non è consistente con i punteggi assegnati alle due squadre. Sono state necessarie due regole in quanto i punteggi delle squadre sono legati alla partita e non direttamente alle squadre.

```
isHomeTeam(?t1, ?m) ^ isAwayTeam(?t2, ?m) ^ hasWinner(?m, ?t1) ^
    MatchHomeTeamScore(?m, ?s1) ^ MatchAwayTeamScore(?m, ?s2) ^
    swrlb:lessThanOrEqual(?s1, ?s2) -> InconsistentMatchScoresError(?m)
```

```
isHomeTeam(?t1, ?m) ^ isAwayTeam(?t2, ?m) ^ hasWinner(?m, ?t2) ^
    MatchHomeTeamScore(?m, ?s1) ^ MatchAwayTeamScore(?m, ?s2) ^
    swrlb:lessThanOrEqual(?s2, ?s1) -> InconsistentMatchScoresError(?m)
```

## 5.11 InvalidLeagueMatchFormatExtraTime/PenaltyRule

Le partite che fanno parte di una lega non possono avere supplementari o rigori, e queste due regole servono a identificare tali inconsistenze.

```
includedInTournament(?m, ?t) ^ League(?t) ^ ExtraTimePlayed(?m, true) ->
    InvalidLeagueMatchFormatError(?m)
```

```
includedInTournament(?m, ?t) ^ League(?t) ^ PenaltyShootoutPlayed(?m,
    true) -> InvalidLeagueMatchFormatError(?m)
```

## 5.12 InvalidMatchRule

Questa regola verifica se una squadra sta giocando contro sé stessa in una partita.

```
hasHomeTeam(?m, ?t) ^ hasAwayTeam(?m, ?t) -> InvalidMatchError(?m)
```

## 5.13 MultipleRedCardsRule

Questa regola controlla se un giocatore ha ricevuto multipli cartellini rossi in una partita.

```
hasReceived(?p, ?c1) ^ hasReceived(?p, ?c2) ^ RedCard(?c1) ^  
  RedCard(?c2) ^ differentFrom(?c1, ?c2) ^ hasEvent(?m, ?c1) ^  
  hasEvent(?m, ?c2) -> MultipleRedCardsError(?p)
```

## 5.14 NonGoalKeeperWithSavesRule

Questa regola verifica se le statistiche di un giocatore indicano il numero di parate sebbene esso non abbia giocato come portiere nella partita.

```
playsInPosition(?pl, ?p) ^ Outfield(?p) ^ hasPlayerStats(?pl, ?ps) ^  
  Saves(?ps, ?s) ^ hasPlayer(?m, ?p) ^ hasPlayerPerformanceStats(?m,  
  ?ps) -> NonGoalKeeperWithSavesError(?pl)
```

## 5.15 PlayerInMatchRule

Questa regola collega un giocatore ad una partita, sapendo che il giocatore era in una formazione utilizzata in tale partita.

```
hasPlayerInFormation(?f, ?p) ^ isFormationIn(?f, ?m) -> hasPlayer(?m, ?p)
```

## 5.16 TeamNotInMatchRule

Questa regola controlla se un evento in una partita è legato ad una squadra che non gioca in tale partita.

```
hasEvent(?m, ?e) ^ hasMatchEvent(?t, ?e) ^ isHomeTeam(?t1, ?m) ^  
  isAwayTeam(?t2, ?m) ^ differentFrom(?t, ?t1) ^ differentFrom(?t, ?t2)  
  -> TeamNotInMatchError(?t)
```

## 5.17 UncontractedPlayerInMatchRule

Questa regola verifica se un giocatore ha partecipato ad una partita senza far parte di una delle due squadre coinvolte, in quanto non si trovava sotto contratto con una delle due squadre al momento della partita. La regola copre soltanto i casi in cui il giocatore si trovava sotto contratto con un'altra squadra e se il contratto è terminato.

```
signs(?t1, ?c) ^ involvesPlayer(?c, ?p) ^ participatesIn(?p, ?m) ^  
  includes(?m, ?t2) ^ includes(?m, ?t3) ^ differentFrom(?t1, ?t2) ^  
  differentFrom(?t1, ?t3) ^ ContractStartDate(?c, ?sd) ^  
  ContractEndDate(?c, ?ed) ^ MatchDate(?m, ?d) ^ temporal:before(?sd,  
  ?d) ^ temporal:after(?ed, ?d) -> UncontractedPlayerInMatchError(?p)
```

## Capitolo 6

# Interrogazioni SPARQL

Capitolo 7

Conclusioni

# Bibliografia

- [1] Mario Rossi. *Citazione di test*. licensed under CC BY 4.0. 2025. URL: <https://example.com>.